
STAGE TICE au service du prof d’EPS 
 

  François Cléaut  juin 13 

 

LA FONCTION RECHERCHE. 
 

 

 

Problème : A partir de la saisie d’une performance, demander au Tableur d’aller chercher 

dans le barème de l’APS, la note correspondante et l’afficher.  

 

Solution : La fonction recherche.  = recherche()   permet par exemple de rechercher 

les valeurs d’un barème. 

 

 

 syntaxeRECHERCHE(valeur saisie; tableau de valeurs similaires; tableau de 

valeurs à afficher) 

 

Ou, adaptée à l'EPS : 

 

 =RECHERCHE ( Performance ;barème ; note ). 

 

La performance =  texte, nombre, test logique. 

 

Le barème     =     C’est une plage de cellules qui permet la comparaison avec la  

performance. 

   Les performances du barème doivent être saisies en ordre croissant  

du haut vers le bas. En revanche l’ordre des notes est indifférent 

mais doit rester bien sur cohérent. 

 

Note    = La fonction recherche renvoie la note correspondant à la performance  

trouvée dans le barème.  

Si la performance saisie se situe entre 2 performances dans le barème,  

la fonction recherche renvoie la valeur correspondant à la performance 

immédiatement inférieure à la performance saisie. 

 

 

 

Créer les outils et écrire la formule. 
 

 Ouvrir le fichier “ recherche.xlsx ” feuille « Hauteur » 

 

 

ETAPE 1 :   Etape générale, sans s’occuper du sexe ni des bornes. 

 

Saisir la formule dans F5 :  =RECHERCHE(E5;H4:H25;I4:I25) 

 

Utilisation de la souris pour sélectionner les plages de barèmes notamment si elles se 

trouvent dans des feuilles ou des classeurs différents. 

Saisir une performance dans E5 et constater. Tester différentes performances, y 

compris extrêmes. 

Les problèmes non résolus : 

 

 Le renvoi de la valeur d’erreur #N/A pour des performances inférieures à 70 cm. 

 La différence garçons – filles n’est pas prise en compte. 

 La recopie reste à faire. 

ETAPE 2 :  Les bornes. 

  

 Si la valeur de l'argument valeur_saisie (perf) est inférieure à la plus petite valeur du 

tableau de valeurs similaires(barème), la fonction RECHERCHE renvoie la valeur 

d'erreur #N/A. 

 En clair, si la performance est inférieure à la plus petite performance existante dans le 

barème, la fonction recherche renvoie la valeur d’erreur #N/A. 

 

 Dans les 2 barèmes ( garçons et filles ), il va simplement  falloir borner les notes 0 

puisque la note 20 ne peut être dépassée ( la fonction recherche renvoie la valeur 

immédiatement inférieure ). 

 Pour cela, saisir une performance = 0 pour une nouvelle note = 0 en haut du tableau. 

 

 
Nous utilisons la plupart du temps des barèmes qui séparent les résultats des 
garçons et ceux des filles. 
 
 

ETAPE 3 :  Utilisation des 2 barèmes garçons et filles. 

 

 Problème à résoudre :  

 Si dans la colonne “  sexe ” est indiqué F, aller chercher la note dans le tableau des 

filles, sinon aller dans le tableau des garçons.  

Une formule conditionnelle doit être créée. 

Fonction conditionnelle 

 

=si() 

 

syntaxe =SI(CONDITION;SI VRAI;SI FAUX) 

 

 1)Condition : c’est la comparaison par l’intermédiaire d’un opérateur de 2 éléments 

 Vrai : Ce qui doit être calculé ou affiché si la condition est vérifiée 

 Faux : Ce qui doit être calculé ou affiché si la condition n’est pas vérifié. 

 

Saisir en F5 : 

  =SI(D5="F";RECHERCHE(E5;H4:H25;I4:I25); 

RECHERCHE(E5;K4:K25;L4:L25)) 
 

 Quand aucune performance n’est saisie dans la colonne E, la note 0 apparaît. 

 Pour éviter cet inconvénient, imbriquer 2 fonctions SI : 

 2)on peut imbriquer des fonctions 

   

 A noter qu’il est souvent intéressant d’écrire en premier lieu sur un papier les formules 

en langage clair, surtout lorsque les formules sont longues et compliquées. Il faut aussi 

être clair dans sa tête, il faut savoir exactement ce que l’on cherche et sous qu’elles 

conditions. 

 

 En fonction de la perf et du sexe on veut calculer la note de l’élève. 

  

 Arbre logique,  
 

    Vrai 1 ‘vide’ 

 Condition 1         vrai 2 recherche/ barème F 
Pas de perf dans la cellule 

    Faux 1  Condition 2 
     Si la cellule contient un F 

        Faux 2  recherche/barème G 

 

 

 

On demande au  tableur de ne renvoyer aucune valeur quand la cellule E5 est vide. 

 

 =SI(E5="";"";SI(D5="F";RECHERCHE(E5;H4:H25;I4:I25) ; 

RECHERCHE(E5;K4:K25;L4:L25))) 
 

Tester puis recopier la formule. Constater. 

 

ETAPE 4 :    Les références relatives et absolues ou nommer les cellules ou plages de  

cellules. 

 

La recopie de la formule écrite en F5 incrémente les références des cellules de la 

colonne E mais également les cellules des barèmes, d’où l’affichage de résultats 

incorrects ou de valeurs d’erreur. 

 

 

2 Solutions pour éviter cette difficulté : 

 

1 .  Utiliser les références relatives, mixtes ou absolues ( $ ) : 

Si une formule saisie en B1 renvoie à une référence de type A1, la recopie 

verticale ( avec la poignée de recopie ) fera apparaître A2 – A3 – A4 … dans les 

formules recopiées en B2 – B3 – B4 … 

 

a  .  Si on souhaite que le recopie verticale ne modifie pas la référence à la cellule   

      A1,  on utilisera la référence mixte   A$1 . Il n’est pas utile de bloquer la  

      référence à la colonne A. 

 

b  .  Si on souhaite que la recopie horizontale fasse toujours référence à la colonne  

       A, on utilisera la référence mixte $A1 . Dans ce cas, une recopie verticale  

       modifiera la référence à la ligne 1. 

c  .   Si on souhaite utiliser une recopie verticale et horizontale faisant référence à  

        la cellule A, on utilisera la référence absolue $A$1 . 

 

Pour la saisie de ces références, on utilisera soit la saisie du symbole $, soit la 

touche F4 . Dans ce dernier cas, ne pas oublier de sélectionner au préalable la zone 

de cellules à modifier. 

 

 

Modifier les références aux plages de cellules des barèmes. La formule devient : 

 

=SI(E5="";"";SI(D5="F";RECHERCHE(E5;H$4:H$25;I$4:I$25) ; 

RECHERCHE(E5;K$4:K$25;L$4:L$25))) 
 

Rappeler l’utilisation de la touche F4. Recopier et constater.  

 

 

 

2 .  Nommer les plages de cellules. 
 

 Sélectionner les plages de cellules des barèmes : 

 

 Sélectionner la colonne de performances en hauteur filles  = Hauteurf 

 Sélectionner la colonne de notes en hauteur filles   = Notef 

Sélectionner la colonne de performances en hauteur garçons = Hauteurg 

Sélectionner la colonne de notes en hauteur garçons  = Noteg 

 

 Reprendre la formule dans la cellule F5, sélectionner la partie  H4:H25 et avec la  

souris sélectionner la zone de performances hauteur filles : Valider et constater.  

Le nom de la zone sélectionné est automatiquement introduit dans la formule.  

Autre solution, sélectionner le nom de plage de cellules voulus dans la zone de noms. 

 

Reproduire cette manipulation pour les 3 autres plages. Recopier et constater.  

Nommer une plage de cellules équivaut à créer une image à laquelle la formule, 

recopiée ou non, fera référence. 

 

La formule devient : 

 

=SI(E5="";"";SI(D5="F";RECHERCHE(E5;Hauteurf;Notef) ; 

RECHERCHE(E5;Hauteurg;Noteg))) 
 

 

 

ETAPE 5 :  Les barèmes de temps.  Ouvrir l’onglet Barèmes 

 

Les temps sont saisis en valeur croissante, c’est à dire du temps le plus court au temps 

le plus long, comme pour la hauteur, de la hauteur la plus basse à la plus haute. 

Mais attention, en EPS un temps faible représente une meilleure performance qu’un 

temps élevé. Il faut donc inverser l’échelle des notes. 

Perf 11.8 12 12.1 

Note attendue 17 16 16 

Note obtenue 17 17 16 

 

 

Le temps 12 est situé entre 11.8 et 12.1 : donc la fonction recherche la valeur 

correspond à la performance immédiatement inférieure, soit la note correspondant à 

11.8. 

 

 

Solution  :  BORNER LES BAREMES. 

 

Pour chaque niveau de note, il faut saisir dans le barème des performances les valeurs 

extrêmes : Par exemple, il faut indiquer que la note 12 est renvoyée pour les 

performances allant de 13.20 à 13.01 incluses. 

 

Construction du barème : 

 

1 .  Saisir les données principales ( cf. [Recherche.xlsx]Barèmes! ) 

2 .  Insérer une ligne sous chacune de ces données 

3 .  Sélectionner une colonne -  Acceuil –Rechercher-- Atteindre – Cellules … – 

Cellules vides 

4 .  Saisir la formule : 

=B4+0.01 CTRL+ENTREE 

Soit prendre rendre B5 immédiatement supérieure à la valeur juste au-dessus en  

ajoutant 0.01". 

On peut réaliser cette opération en 1 seule fois en sélectionnant tous les 

barèmes de performances. 

5 .  Reproduire la même opération pour les notes avec la formule : 

=C4  CTRL+ENTREE 

Soit reproduire la note immédiatement inférieure. 

6 .  Vous pouvez ajouter une ligne pour les élèves dispensés. 

 

 

 

L'intérêt de cette solution est de pouvoir utiliser la même formule que pour les 

épreuves mesurées, et de pouvoir l'insérer dans une même feuille. Elle permet 

également d'intégrer dans les barèmes toutes les informations alphanumériques. 

 

 


